

This is an introduction to MAPLE which is a program for doing mathematics with computers (a computer algebra system).

We will focus on permutations, groups, and discrete mathematics. But you certainly would benefit from learning more about this wonderful tool!

A PERMUTATION inside MAPLE can be represented in two ways:

(1) A permutation list (1->3,2->1,3->2) which is encoded by a list [3,1,2].

(2) As the product of disjoint cycles: (1,2)(3,2,4) is encoded in a list of lists [[1,2],[3,2,4]].

We can construct many groups using permutations. In fact, ALL finite groups can be constructed using permutations! Here is an example:

We have talked about the cheap_rubik box. Instead of a 3x3x3 cube it is a 3x3x2 box. The point is that once we label the faces of the cheap_rubik box we can express the mechanical moves by a series of permutations of 18 symbols!!

The subgroup of S_{18} generated by these permutations is precisely the group of operations for the cheap_rubik box. Now here is a practical problem. I give you a configuration that could in principle be a valid cheap_rubik configuration. How can we decide whether there are moves that really achieve it?

Take for example [10,6,9,11,5,8,12,4,7,3,2,1,13,14,15,16,17,18]

Let's begin our computer exploration:

Since we are beginners we can just ask for help:

```
> help(group);
```

Now we call the library of commands that deals with groups and permutations.

```
> with(group);
```

```
[DerivedS, LCS, NormalClosure, RandElement, SnConjugates, Sylow, areconjugate, center,
  centralizer, core, cosets, cosrep, derived, elements, groupmember, grouporder, inter, invperm,
  isabelian, isnormal, issubgroup, mulperms, normalizer, orbit, parity, permrep, pres, transgroup]
```

Let's now define the moves of the cheap rubik box.

```
> moves_rubik:={
  [[1, 3, 9, 7], [2, 6, 8, 4]],
  [[10, 12, 18, 16], [11, 15, 17, 13]],
  [[1, 12], [3, 10], [2, 11]],
  [[7, 18], [9, 16], [8, 17]],
```

```

[[1,16],[4,13],[7,10]],
[[3,18],[6,15],[9,12]]};
moves_rubik := {[[1,3,9,7],[2,6,8,4]],[[10,12,18,16],[11,15,17,13]],
[[1,12],[3,10],[2,11]],[[7,18],[9,16],[8,17]],[[1,16],[4,13],[7,10]],
[[3,18],[6,15],[9,12]]}
[ The group is generated by these moves:
> cheaprubik:=permgrou(18,moves_rubik);
cheaprubik:=permgrou(18, {[[1,3,9,7],[2,6,8,4]],[[10,12,18,16],[11,15,17,13]],
[[1,12],[3,10],[2,11]],[[7,18],[9,16],[8,17]],[[1,16],[4,13],[7,10]],
[[3,18],[6,15],[9,12]]})
[ Then we can ask questions about the group:
How many elements does it have?
> grouporder(cheaprubik);
1625702400
[ Is it abelian?
> isabelian(cheaprubik);
false
[ You might want to convert between the list and cycle representation of permutations:
> b:=convert([10,6,9,11,5,8,12,4,7,3,2,1,13,14,15,16,17,18], 'disjcyc
');
b := [[1,10,3,9,7,12],[2,6,8,4,11]]
> convert(b, 'permlist', 18);
[10,6,9,11,5,8,12,4,7,3,2,1,13,14,15,16,17,18]
[ Is a particular permutation in the group?
> groupmember([[1,10,3,9,7,12],[2,6,8,4,11]], cheaprubik);
true
> groupmember([[1,10,11]], cheaprubik);
false
> groupmember([[1,10,3,9,7,12],[2,6,8,4,11]], cheaprubik);
true
[ Here is a way to multiply two permutations as long as they are in disjoint
cycle notation (careful about the order!!):
> mulperms([[1,2]],[[1,2,3]]);
[[1,3]]
[ Now suppose you wish to decide what are the cosets of a certain subgroup
(like in the proof of the Lagrange theorem). First you input the group and
the candidate subgroup:
> pg:=permgrou(4, {
[[1,2,3]],[[1,3,2]],
[[1,3,4]],[[1,4,3]],
[[1,2,4]],[[1,4,2]],
[[2,3,4]],[[2,4,3]]});
pg := permgrou(4,
[[[1,3,2]],[[1,3,4]],[[1,4,3]],[[1,2,4]],[[1,4,2]],[[2,3,4]],[[2,4,3]],[[1,2,3]]])

```

```

[ > grouporder(pg);
                                12
[ We might want to list all elements in the group:
[ > elements(pg);
{ [], [[1, 3, 2]], [[1, 3, 4]], [[1, 4, 3]], [[1, 2, 4]], [[1, 4, 2]], [[2, 3, 4]], [[2, 4, 3]], [[1, 2, 3]],
  [[1, 4], [2, 3]], [[1, 3], [2, 4]], [[1, 2], [3, 4]]}
[ As you can see this is in fact  $A_4$ , the set of all even permutations on 4 letters!!
[ > sg:=permgrou(4, {[ [1, 2], [3, 4] ]});
                                sg := permgrou(4, {[ [1, 2], [3, 4] ]})
[ > grouporder(sg);
                                2
[ Let's first make sure that sub is indeed a subgroup of pg:
[ > issubgroup(sg,pg);
                                true
[ Finally, we ask for the list of cosets to be computed. The answer is a list of coset representatives:
[ > cosets(pg,sg);
{ [], [[1, 2, 4]], [[2, 3, 4]], [[2, 4, 3]], [[1, 2, 3]], [[1, 3], [2, 4]]}
[ This gives the number of inequivalent configurations of the cheap rubik box
(two configurations are inequivalent, if there is no sequence of simple moves which brings one to the
other):
[ > 18!/grouporder(cheaprubik);
                                3938220
[ Now let's determine the orbits of the facets of the cheap rubik box!
[ To make our lifes easier let's first define a function:
[ > f:=n->orbit(cheaprubik,n);
[ The orbit of a corner:
[ > f(1);
                                {1, 3, 7, 9, 10, 12, 16, 18}
[ The orbit of an edge:
[ > f(2);
                                {2, 4, 6, 8, 11, 13, 15, 17}
[ The orbit of the center pieces:
[ > f(5);f(14);
                                {5}
                                {14}
[ What is the order of the subgroup of cheaprubik generated by moves of order 2?
[ > moves_sg:={
  [[1,12],[3,10],[2,11]],
  [[7,18],[9,16],[8,17]],
  [[1,16],[4,13],[7,10]],
  [[3,18],[6,15],[9,12]]}:
  cheap_sg:=permgrou(18,moves_sg):
  grouporder(cheap_sg);
                                192
[ What are the orbit of this subgroup?
[

```

```
[ > g:=n->orbit(cheap_sg,n):  
[ > g(1);  
                                {1, 9, 12, 16}  
[ > g(2);  
                                {2, 11}  
[ > g(8);  
                                {8, 17}  
[ > g(7);  
                                {3, 7, 10, 18}  
[ > g(4);  
                                {4, 13}  
[ > g(6);  
                                {6, 15}  
[ > g(5);  
                                {5}
```